

Homework #4-SOLUTIONS

1. Reading a Space-Delimited Raw Data File

```
data work.qtrdonation;
  length IDNum $ 6;
  infile "&path\donation.dat";
  input IDNum $ Qtr1 Qtr2 Qtr3 Qtr4;
run;

proc print data=work.qtrdonation;
run;
```

2. Reading a Delimited Raw Data File with Nonstandard Values

```
data work.prices;
  infile "&path\pricing.dat" dlm='*';
  input ProductID StartDate :date. EndDate :date.
        Cost :dollar. SalesPrice :dollar.;
  format StartDate EndDate mmddyy10.
        Cost SalesPrice 8.2;
run;

title '2011 Pricing';
proc print data=work.prices;
run;
title;
```

3. Reading a Delimited File with Missing Values

```
data work.prices;
  infile "&path\prices.dat" dlm='*' missover;
  input ProductID StartDate :date. EndDate :date.
        UnitCostPrice :dollar. UnitSalesPrice :dollar.;
  label ProductID='Product ID'
        StartDate='Start of Date Range'
        EndDate='End of Date Range'
        UnitCostPrice='Cost Price per Unit'
        UnitSalesPrice='Sales Price per Unit';
  format StartDate EndDate mmddyy10.
        UnitCostPrice UnitSalesPrice 8.2;
run;

title '2007 Prices';
proc print data=work.prices label;
run;
title;
```

4. Creating New Variables

```
data work.birthday;
  set orion.customer;
  Bday2012=mdy(month(Birth_Date),day(Birth_Date),2012);
  BdayDOW2012=weekday(Bday2012);
  Age2012=(Bday2012-Birth_Date)/365.25;
  keep Customer_Name Birth_Date Bday2012 BdayDOW2012 Age2012;
```

```

format Bday2012 date9. Age2012 3.;
run;

proc print data=work.birthday;
run;

```

5. Creating Multiple Variables in Conditional Processing

```

data work.season;
  set orion.customer_dim;
  length Promo2 $ 6;
  Quarter=qtr(Customer_BirthDate);
  if Quarter=1 then Promo='Winter';
  else if Quarter=2 then Promo='Spring';
  else if Quarter=3 then Promo='Summer';
  else if Quarter=4 then Promo='Fall';
  if Customer_Age>=18 and Customer_Age<=25 then Promo2='YA';
  else if Customer_Age>=65 then Promo2='Senior';
  keep Customer_FirstName Customer_LastName Customer_BirthDate
        Customer_Age Promo Promo2;
run;

proc print data=work.season;
  var Customer_FirstName Customer_LastName Customer_BirthDate Promo
        Customer_Age Promo2;
run;

```

6. Creating Variables Unconditionally and Conditionally

```

data work.ordertype;
  set orion.orders;
  length Type $ 13 SaleAds $ 5;
  DayOfWeek=weekday(Order_Date);
  if Order_Type=1 then
    Type='Retail Sale';
  else if Order_Type=2 then do;
    Type='Catalog Sale';
    SaleAds='Mail';
  end;
  else if Order_Type=3 then do;
    Type='Internet Sale';
    SaleAds='Email';
  end;
  drop Order_Type Employee_ID Customer_ID;
run;

proc print data=work.ordertype;
run;

```

7. Reading a Tab-Delimited Raw Data File

```

data work.managers2;
  length First Last $ 12 Title $ 25;
  infile "&path\managers2.dat" dlm='09'x;
  input ID First $ Last $ Gender $ Salary Title $;
  keep First Last Title;
run;

proc print data=work.managers2;
run;

```

8. Reading a Delimited File with Missing Values and Embedded Delimiters

```
data work.salesmgmt;
  length First Last $ 12 Gender $ 1 Title $ 25 Country $ 2;
  format BirthDate HireDate date9.;
  infile "&path\managers.dat" dsd dlm='/' missover;
  input ID First Last Gender Salary Title Country
        BirthDate :date. HireDate :mmddyy.;
run;

title 'Orion Star Managers';
proc print data=work.salesmgmt;
  var ID Last Title HireDate Salary;
run;
title;

/* Alternate solution using informats */
data work.salesmgmt;
  format BirthDate HireDate date9.;
  infile "&path\managers.dat" dsd dlm='/' missover;
  input ID First :$12. Last :$12. Gender :$1. Salary Title :$25.
        Country :$2. BirthDate :date. HireDate :mmddyy.;
run;
```

9. Using the CATX and INTCK Functions to Create Variables

```
data work.employees;
  set orion.sales;
  FullName=catx(' ',First_Name,Last_Name);
  Yrs2012=intck('year',Hire_Date,'01JAN2012'd);
  format Hire_Date ddmmyy10.;
  label Yrs2012='Years of Employment in 2012';
run;

proc print data=work.employees label;
  var FullName Hire_Date Yrs2012;
run;
```

10. Using WHEN Statements in a SELECT Group to Create Variables Conditionally

```
data work.gifts;
  set orion.nonsales;
  length Gift1 $ 6 Gift2 $ 10;
  select(Gender);
    when('F') do;
      Gift1='Scarf';
      Gift2='Pedometer';
    end;
    when('M') do;
      Gift1='Gloves';
      Gift2='Money Clip';
    end;
    otherwise do;
      Gift1='Coffee';
      Gift2='Calendar';
    end;
  end;
  keep Employee_ID First Last Gender Gift1 Gift2;
run;
```

```
proc print data=work.gifts noobs;  
run;
```